

Brand Awareness Study

A major Canadian coffee shop chain is trying to break into the U.S. Market. They assess the following variables twice on a random sample of coffee-drinking adults. The two measurements of each variable are conducted at different times by different interviewers asking somewhat different questions, in such a way that the errors of measurement may be assumed independent. The variables are

Brand Awareness: Familiarity with the coffee shop chain


Advertising Awareness: Recall for advertising of the coffee shop chain

Interest in the product category: Mostly this was how much they say they like doughnuts.

Purchase Intention: Expressed willingness to go to an outlet of the coffeeshop chain and make an order.

Purchase behaviour: Reported dollars spent at the chain during the 2 months following the interview.

All variables were measured on a scale from 0 to 100 except purchase behaviour, which is in dollars.



```

/* aware1.sas */
options linesize=79 pagesize = 500 noovp formdlim='_';
title 'Doughnut Shop Brand Awareness Study: Model 1';

data torus;
  infile 'timmy1.data' firstobs=2;
  input id w1 w2 w3 w4 w5 w6 v1 v2 v3 v4;
  label w1 = 'Brand Awareness 1'
        w2 = 'Brand Awareness 2'
        w3 = 'Ad Awareness 1'
        w4 = 'Ad Awareness 2'
        w5 = 'Interest 1'
        w6 = 'Interest 2'
        v1 = 'Purchase Intention 1'
        v2 = 'Purchase Intention 2'
        v3 = 'Purchase Behaviour 1'
        v4 = 'Purchase Behaviour 2';

ods exclude Calis.ML.SqMultCorr (persist);

/* F_BrAw = True brand awareness
   F_AdAw = True advertising awareness
   F_Inter = True interest in the product category
   F_PI = True purchase intention
   F_PBeh = True purchase behaviour */

proc calis psummary; /* Print fit statistics only */
  var w1 w2 w3 w4 w5 w6 v1 v2 v3 v4;
  linesq
    F_PI = gamma1 F_BrAw + gamma2 F_AdAw
          + gamma3 F_Inter + epsilon1,
    F_PBeh = gamma4 F_Inter + beta F_PI + epsilon2,
    w1 = F_BrAw + e1,
    w2 = F_BrAw + e2,
    w3 = F_AdAw + e3,
    w4 = F_AdAw + e4,
    w5 = F_Inter + e5,
    w6 = F_Inter + e6,
    v1 = F_PI + e7,
    v2 = F_PI + e8,
    v3 = F_PBeh + e9,
    v4 = F_PBeh + e10;
  variance
    F_BrAw = phi11, F_AdAw=phi22, F_Inter=phi33,
    epsilon1=psi1, epsilon2=psi2,
    e1-e10 = 10 * omega__;
  cov
    F_BrAw F_AdAw = phi12, F_BrAw F_Inter = phi13,
    F_AdAw F_Inter = phi23;
  bounds phi11 phi22 phi33 psi1 psi2 omega01-omega10 > 0;


```

The CALIS Procedure
Covariance Structure Analysis: Maximum Likelihood Estimation

Fit Summary

Modeling Info	N Observations	200
	N Variables	10
	N Moments	55
	N Parameters	23
	N Active Constraints	0
	Baseline Model Function Value	4.9657
	Baseline Model Chi-Square	988.1793
	Baseline Model Chi-Square DF	45
	Pr > Baseline Model Chi-Square	<.0001
	Absolute Index	Fit Function
Chi-Square		77.3629
Chi-Square DF		32
Pr > Chi-Square		<.0001
Z-Test of Wilson & Hilferty		4.1889
Hoelter Critical N		119
Root Mean Square Residual (RMSR)		2.6711
Standardized RMSR (SRMSR)		0.1228
Goodness of Fit Index (GFI)		0.9263
Parsimony Index		Adjusted GFI (AGFI)
	Parsimonious GFI	0.6587
	RMSEA Estimate	0.0844
	RMSEA Lower 90% Confidence Limit	0.0606
	RMSEA Upper 90% Confidence Limit	0.1086
	Probability of Close Fit	0.0108
	ECVI Estimate	0.6334
	ECVI Lower 90% Confidence Limit	0.5213
	ECVI Upper 90% Confidence Limit	0.7867
	Akaike Information Criterion	123.3629
Incremental Index	Bozdogan CAIC	222.2242
	Schwarz Bayesian Criterion	199.2242
	McDonald Centrality	0.8928
	Bentler Comparative Fit Index	0.9519
	Bentler-Bonett NFI	0.9217
	Bentler-Bonett Non-normed Index	0.9324
	Bollen Normed Index Rho1	0.8899
	Bollen Non-normed Index Delta2	0.9526
	James et al. Parsimonious NFI	0.6554

Split the problem up into parts. Look just at the measurement model.



```

/* aware2.sas */
options linesize=79 pagesize = 500 noovp formdlim='_';
title 'Doughnut Shop Brand Awareness Study: Model 2 (Measurement)';

data torus;
infile 'timmy1.data' firstobs=2;
input id w1 w2 w3 w4 w5 w6 v1 v2 v3 v4;
label w1 = 'Brand Awareness 1'
      w2 = 'Brand Awareness 2'
      w3 = 'Ad Awareness 1'
      w4 = 'Ad Awareness 2'
      w5 = 'Interest 1'
      w6 = 'Interest 2'
      v1 = 'Purchase Intention 1'
      v2 = 'Purchase Intention 2'
      v3 = 'Purchase Behaviour 1'
      v4 = 'Purchase Behaviour 2';

ods exclude Calis.ML.SqMultCorr (persist);

proc calis psummary; /* Print fit statistics only */
var w1 w2 w3 w4 w5 w6 v1 v2 v3 v4;
lineqs
w1 = F_BrAw + e1,
w2 = F_BrAw + e2,
w3 = F_AdAw + e3,
w4 = F_AdAw + e4,
w5 = F_Inter + e5,
w6 = F_Inter + e6,
v1 = F_PI + e7,
v2 = F_PI + e8,
v3 = F_PBeh + e9,
v4 = F_PBeh + e10;
variance
F_BrAw F_AdAw F_Inter F_PI F_PBeh = 5*phivar__,
e1-e10 = 10 * omega__;
cov F_BrAw F_AdAw F_Inter F_PI F_PBeh = 10*phicov__;
bounds phivar1-phivar5 omega01-omega10 > 0;

```

Chi-Square	75.9983
Chi-Square DF	30
Pr > Chi-Square	<.0001

The measurement model does not fit. Try a true double measurement model, allowing covariances within sets.

```

/* aware3.sas */
options linesize=79 pagesize = 500 noovp formdlim='_';
title 'Doughnut Shop Brand Awareness Study';
title2 'Full double measuremntn model';

data torus;
  infile 'timmy1.data' firstobs=2;
  input id w1 w2 w3 w4 w5 w6 v1 v2 v3 v4;
  label w1 = 'Brand Awareness 1'
        w2 = 'Brand Awareness 2'
        w3 = 'Ad Awareness 1'
        w4 = 'Ad Awareness 2'
        w5 = 'Interest 1'
        w6 = 'Interest 2'
        v1 = 'Purchase Intention 1'
        v2 = 'Purchase Intention 2'
        v3 = 'Purchase Behaviour 1'
        v4 = 'Purchase Behaviour 2';

ods exclude Calis.ML.SqMultCorr (persist);

proc calis psummary; /* Print fit statistics only */
  var w1 w2 w3 w4 w5 w6 v1 v2 v3 v4;
  lineqs
    w1 = F_BrAw + e1,
    w2 = F_BrAw + e2,
    w3 = F_AdAw + e3,
    w4 = F_AdAw + e4,
    w5 = F_Inter + e5,
    w6 = F_Inter + e6,
    v1 = F_PI + e7,
    v2 = F_PI + e8,
    v3 = F_PBeh + e9,
    v4 = F_PBeh + e10;
  variance
    F_BrAw F_AdAw F_Inter F_PI F_PBeh = 5*phivar__,
    e1-e10 = 10 * omega__;
  cov F_BrAw F_AdAw F_Inter F_PI F_PBeh = 10*phicov__,
    e1 e3 e5 e7 e9 = 10 * O1cov__,
    e2 e4 e6 e8 e10 = 10 * O2cov__;
  bounds phivar1-phivar5 omega01-omega10 > 0;

```

Chi-Square	33.1432
Chi-Square DF	10
Pr > Chi-Square	0.0003

Also, get this warning in the log file:


WARNING: Although all predicted variances for the observed and latent variables are positive, the corresponding predicted covariance matrix is not positive definite. It has one negative eigenvalue.

Testing the covariances with an invalid likelihood ratio test:

```
> G2 = 75.9983 - 33.1432; G2
[1] 42.8551
> 1-pchisq(G2,df=20)
[1] 0.00213638
```

This shows why you should not believe a LR based on a full model that does not fit. Later we will conclude that those covariances are truly zero.

Consider that the two measurements of each latent variable are DIFFERENT. Maybe they're not really equivalent. Perhaps one in each set (say number 2) should have a coefficient not equal to one.




Aware4.sas is like aware3.sas except for this change in the model equations:

```
lineqs
w1 = F_BrAw + e1,
w2 = lambda2*F_BrAw + e2,
w3 = F_AdAw + e3,
w4 = lambda4*F_AdAw + e4,
w5 = F_Inter + e5,
w6 = lambda6*F_Inter + e6,
v1 = F_PI + e7,
v2 = lambda8*F_PI + e8,
v3 = F_PBeh + e9,
v4 = lambda10*F_PBeh + e10;
```

It fits!

Chi-Square	17.7481
Chi-Square DF	25
Pr > Chi-Square	0.8529

Now combine this with the latent variable model.



```

/* aware5.sas */
options linesize=79 pagesize = 500 noovp formdlim='_';
title 'Doughnut Shop Brand Awareness Study: Model 5';
title2 'Put latent variable model back in';

data torus;
  infile 'timmy1.data' firstobs=2;
  input id w1 w2 w3 w4 w5 w6 v1 v2 v3 v4;
  label w1 = 'Brand Awareness 1'
        w2 = 'Brand Awareness 2'
        w3 = 'Ad Awareness 1'
        w4 = 'Ad Awareness 2'
        w5 = 'Interest 1'
        w6 = 'Interest 2'
        v1 = 'Purchase Intention 1'
        v2 = 'Purchase Intention 2'
        v3 = 'Purchase Behaviour 1'
        v4 = 'Purchase Behaviour 2';

ods exclude Calis.ML.SqMultCorr (persist);

/* F_BrAw = True brand awareness
   F_AdAw = True advertising awareness
   F_Inter = True interest in the product category
   F_PI = True purchase intention
   F_PBeh = True purchase behaviour */

proc calis pshort nostand pcorr;
  var w1 w2 w3 w4 w5 w6 v1 v2 v3 v4;
  lineqs
    F_PI = gamma1 F_BrAw + gamma2 F_AdAw
          + gamma3 F_Inter + epsilon1,
    F_PBeh = gamma4 F_Inter + beta F_PI + epsilon2,
    w1 = F_BrAw + e1,
    w2 = lambda2*F_BrAw + e2,
    w3 = F_AdAw + e3,
    w4 = lambda4*F_AdAw + e4,
    w5 = F_Inter + e5,
    w6 = lambda6*F_Inter + e6,
    v1 = F_PI + e7,
    v2 = lambda8*F_PI + e8,
    v3 = F_PBeh + e9,
    v4 = lambda10*F_PBeh + e10;
  variance
    F_BrAw = phi11, F_AdAw=phi22, F_Inter=phi33,
    epsilon1=psi1, epsilon2=psi2,
    e1-e10 = 10 * omega__;
  cov
    F_BrAw F_AdAw = phi12, F_BrAw F_Inter = phi13,
    F_AdAw F_Inter = phi23;
  bounds phi11 phi22 phi33 psi1 psi2 omega01-omega10 > 0;

/* The lambda coefficients linking latent variables to their measurements
are different from zero. They'd better be! But which of them are different
from one? */
  simtests not_one = [d2 d4 d6 d8 d10];
    d2 = lambda2-1; d4 = lambda4-1; d6 = lambda6-1;
    d8 = lambda8-1; d10 = lambda10-1;

```


Doughnut Shop Brand Awareness Study: Model 5
Put latent variable model back in

1

The CALIS Procedure
Covariance Structure Analysis: Model and Initial Values

Modeling Information

Data Set	WORK.TORUS
N Records Read	200
N Records Used	200
N Obs	200
Model Type	LINEQS
Analysis	Covariances

Doughnut Shop Brand Awareness Study: Model 5
Put latent variable model back in

2

The CALIS Procedure
Covariance Structure Analysis: Descriptive Statistics

Covariance Matrix (DF = 199)

	w1	w2	w3	w4	w5
w1 Brand Awareness 1	24.36259	10.18613	12.38641	6.67043	13.12611
w2 Brand Awareness 2	10.18613	18.41568	6.65508	2.89307	7.62794
w3 Ad Awareness 1	12.38641	6.65508	23.06309	8.68460	10.62490
w4 Ad Awareness 2	6.67043	2.89307	8.68460	18.18389	6.53686
w5 Interest 1	13.12611	7.62794	10.62490	6.53686	21.29224
w6 Interest 2	15.12312	7.76884	12.76633	7.39950	16.40955
v1 Purchase Intention 1	16.20631	8.34794	14.88942	7.56339	16.43314
v2 Purchase Intention 2	11.92862	5.88221	11.33997	6.36148	10.88168
v3 Purchase Behaviour 1	7.83822	4.67457	5.66204	3.73802	6.14545
v4 Purchase Behaviour 2	8.16269	4.22111	6.89384	2.80339	6.29083

Covariance Matrix (DF = 199)

	w6	v1	v2	v3	v4
w1 Brand Awareness 1	15.12312	16.20631	11.92862	7.83822	8.16269
w2 Brand Awareness 2	7.76884	8.34794	5.88221	4.67457	4.22111
w3 Ad Awareness 1	12.76633	14.88942	11.33997	5.66204	6.89384
w4 Ad Awareness 2	7.39950	7.56339	6.36148	3.73802	2.80339
w5 Interest 1	16.40955	16.43314	10.88168	6.14545	6.29083
w6 Interest 2	24.05025	16.89196	11.25377	7.65075	7.69095
v1 Purchase Intention 1	16.89196	29.59776	15.02460	9.58646	10.11997
v2 Purchase Intention 2	11.25377	15.02460	20.98369	6.99822	6.73053
v3 Purchase Behaviour 1	7.65075	9.58646	6.99822	21.70329	17.74686
v4 Purchase Behaviour 2	7.69095	10.11997	6.73053	17.74686	22.28078

Determinant 201774472804 Ln 26.030416

Doughnut Shop Brand Awareness Study: Model 5
Put latent variable model back in

3

The CALIS Procedure
Covariance Structure Analysis: Optimization

Levenberg-Marquardt Optimization

Scaling Update of More (1978)

Parameter Estimates 28
 Functions (Observations) 55
 Lower Bounds 15
 Upper Bounds 0

Optimization Start

Active Constraints 0 Objective Function 0.1016760046
 Max Abs Gradient Element 0.0313316908 Radius 1

Iter	Rest arts	Func Calls	Act Con	Objective Function	Obj Fun Change	Max Abs Gradient Element	Lambda	Actual Over Pred Change
1	0	4	0	0.09503	0.00665	0.00853	0	0.928
2	0	6	0	0.09482	0.000210	0.000876	0	0.932
3	0	8	0	0.09481	3.935E-6	0.000242	0	0.795
4	0	10	0	0.09481	2.198E-7	0.000052	0	0.752
5	0	12	0	0.09481	1.484E-8	0.000014	0	0.737
6	0	14	0	0.09481	1.062E-9	3.543E-6	0	0.732

Optimization Results

Iterations 6 Function Calls 17
 Jacobian Calls 8 Active Constraints 0
 Objective Function 0.0948118648 Max Abs Gradient Element 3.5432106E-6
 Lambda 0 Actual Over Pred Change 0.7317126267
 Radius 0.000356371

Convergence criterion (ABSGCONV=0.00001) satisfied.

Doughnut Shop Brand Awareness Study: Model 5
 Put latent variable model back in

4

The CALIS Procedure
 Covariance Structure Analysis: Maximum Likelihood Estimation

Fit Summary

Modeling Info	N Observations	200
	N Variables	10
	N Moments	55
	N Parameters	28
	N Active Constraints	0
	Baseline Model Function Value	4.9657
	Baseline Model Chi-Square	988.1793
	Baseline Model Chi-Square DF	45
	Pr > Baseline Model Chi-Square	<.0001
	Absolute Index	Fit Function
Chi-Square		18.8676
Chi-Square DF		27
Pr > Chi-Square		0.8748
Z-Test of Wilson & Hilferty		-1.1505
Hoelter Critical N		424
Parsimony Index	Root Mean Square Residual (RMSR)	0.4214
	Standardized RMSR (SRMSR)	0.0195
	Goodness of Fit Index (GFI)	0.9822
	Adjusted GFI (AGFI)	0.9638

	Parsimonious GFI	0.5893
	RMSEA Estimate	0.0000
	RMSEA Lower 90% Confidence Limit	0.0000
	RMSEA Upper 90% Confidence Limit	0.0286
	Probability of Close Fit	0.9939
	ECVI Estimate	0.3927
	ECVI Lower 90% Confidence Limit	0.4415
	ECVI Upper 90% Confidence Limit	0.4561
	Akaike Information Criterion	74.8676
	Bozdogan CAIC	195.2204
	Schwarz Bayesian Criterion	167.2204
	McDonald Centrality	1.0205
Incremental Index	Bentler Comparative Fit Index	1.0000
	Bentler-Bonett NFI	0.9809
	Bentler-Bonett Non-normed Index	1.0144
	Bollen Normed Index Rho1	0.9682
	Bollen Non-normed Index Delta2	1.0085
	James et al. Parsimonious NFI	0.5885

Predicted Covariances

		w1	w2	w3	w4	w5
w1	Brand Awareness 1	24.36259	10.18613	12.36235	6.71424	13.54737
w2	Brand Awareness 2	10.18613	18.41568	6.52570	3.54424	7.15124
w3	Ad Awareness 1	12.36235	6.52570	23.06309	8.68460	11.36898
w4	Ad Awareness 2	6.71424	3.54424	8.68460	18.18389	6.17472
w5	Interest 1	13.54737	7.15124	11.36898	6.17472	21.29224
w6	Interest 2	14.79014	7.80726	12.41191	6.74116	16.41594
v1	Purchase Intention 1	16.47948	8.69901	15.02103	8.15822	15.61774
v2	Purchase Intention 2	11.65002	6.14969	10.61898	5.76738	11.04081
v3	Purchase Behaviour 1	7.24327	3.82350	6.72879	3.65454	6.58054
v4	Purchase Behaviour 2	7.53115	3.97546	6.99622	3.79978	6.84207

Predicted Covariances

		w6	v1	v2	v3	v4
w1	Brand Awareness 1	14.79014	16.47948	11.65002	7.24327	7.53115
w2	Brand Awareness 2	7.80726	8.69901	6.14969	3.82350	3.97546
w3	Ad Awareness 1	12.41191	15.02103	10.61898	6.72879	6.99622
w4	Ad Awareness 2	6.74116	8.15822	5.76738	3.65454	3.79978
w5	Interest 1	16.41594	15.61774	11.04081	6.58054	6.84207
w6	Interest 2	24.05025	17.05043	12.05365	7.18421	7.46973
v1	Purchase Intention 1	17.05043	29.59776	15.04085	9.59347	9.97475
v2	Purchase Intention 2	12.05365	15.04085	20.98369	6.78202	7.05156
v3	Purchase Behaviour 1	7.18421	9.59347	6.78202	21.70329	17.74686
v4	Purchase Behaviour 2	7.46973	9.97475	7.05156	17.74686	22.28078

Determinant 221841345845 Ln 26.125228


The CALIS Procedure
 Covariance Structure Analysis: Maximum Likelihood Estimation

Linear Equations

$$\begin{aligned}
 F_PI &= 0.2295 * F_BrAw + 0.3687 * F_AdAw + 0.5531 * F_Inter \\
 Std\ Err & 0.1455 \text{ gamma1} \quad 0.1617 \text{ gamma2} \quad 0.1705 \text{ gamma3} \\
 t\ Value & 1.5772 \quad 2.2798 \quad 3.2446 \\
 & + 1.0000 \text{ epsilon1}
 \end{aligned}$$

Linear Equations

$$\begin{aligned}
 F_PBeh &= -0.1292 * F_Inter + 0.5458 * F_PI + 1.0000 \text{ epsilon2} \\
 Std\ Err & 0.2579 \text{ gamma4} \quad 0.2245 \text{ beta} \\
 t\ Value & -0.5010 \quad 2.4315
 \end{aligned}$$



Linear Equations

$$w1 = 1.0000 F_{BrAw} + 1.0000 e1$$

Linear Equations

$$w2 = 0.5279 * F_{BrAw} + 1.0000 e2$$

Std Err 0.0771 λ^2
t Value 6.8442

Linear Equations

$$w3 = 1.0000 F_{AdAw} + 1.0000 e3$$

Linear Equations

$$w4 = 0.5431 * F_{AdAw} + 1.0000 e4$$

Std Err 0.0906 λ^4
t Value 5.9979

Linear Equations

$$w5 = 1.0000 F_{Inter} + 1.0000 e5$$

Linear Equations

$$w6 = 1.0917 * F_{Inter} + 1.0000 e6$$

Std Err 0.0809 λ^6
t Value 13.4937

Linear Equations

$$v1 = 1.0000 F_{PI} + 1.0000 e7$$

Linear Equations

$$v2 = 0.7069 * F_{PI} + 1.0000 e8$$

Std Err 0.0660 λ^8
t Value 10.7176

Linear Equations

$$v3 = 1.0000 F_{PBeh} + 1.0000 e9$$

Linear Equations

$$v4 = 1.0397 * F_{PBeh} + 1.0000 e10$$

Std Err 0.1102 λ^{10}
t Value 9.4331

Estimates for Variances of Exogenous Variables

Variable Type	Variable	Parameter	Estimate	Standard Error	t Value
Latent	F_BrAw	phi11	19.29669	3.13352	6.15814
	F_AdAw	phi22	15.99021	3.05552	5.23321
	F_Inter	phi33	15.03655	2.16915	6.93199
Disturbance	epsilon1	psi1	3.31743	1.35011	2.45715
	epsilon2	psi2	12.68309	2.11239	6.00413
Error	e1	omega01	5.06590	2.09037	2.42345
	e2	omega02	13.03873	1.42404	9.15619
	e3	omega03	7.07288	2.23517	3.16436
	e4	omega04	13.46712	1.48786	9.05136
	e5	omega05	6.25568	0.96719	6.46789
	e6	omega06	6.12839	1.07124	5.72085
	e7	omega07	8.32179	1.49034	5.58381
	e8	omega08	10.35070	1.22403	8.45626
	e9	omega09	4.63480	1.69444	2.73529
	e10	omega10	3.82859	1.80292	2.12355

Covariances Among Exogenous Variables

Var1	Var2	Parameter	Estimate	Standard Error	t Value
F_BrAw	F_AdAw	phi12	12.36235	1.87827	6.58177
F_BrAw	F_Inter	phi13	13.54737	1.84523	7.34185
F_AdAw	F_Inter	phi23	11.36898	1.70653	6.66205

From the program file:

```
/* The lambda coefficients linking latent variables to their measurements
are different from zero. They'd better be! But which of them are different
from one? */
```

```
simtests not_one = [d2 d4 d6 d8 d10];
d2 = lambda2-1; d4 = lambda4-1; d6 = lambda6-1;
d8 = lambda8-1; d10 = lambda10-1;
```

Simultaneous Tests

Simultaneous Test	Parametric Function	Function Value	DF	Chi-Square	p Value
not_one			5	84.08457	<.0001
	d2	-0.47213	1	37.47264	<.0001
	d4	-0.45688	1	25.45741	<.0001
	d6	0.09174	1	1.28558	0.2569
	d8	-0.29306	1	19.73968	<.0001
	d10	0.03974	1	0.13001	0.7184

Can get these tests by hand from the printout, testing $H_0: \lambda_j = 1$

For example, λ_{10} :

```
v4      = 1.0397*F_PBeh + 1.0000 e10
Std Err 0.1102 lambda10
t Value  9.4331
```

```
> z = (1.0397-1)/0.1102; z^2
```

```
[1] 0.129783
```

```
> 2*(1-pnorm(z))
```

```
[1] 0.7186571
```

```

/* aware6.sas */
options linesize=79 pagesize = 500 noovp formdlim=' ' ;
title 'Doughnut Shop Brand Awareness Study: Model 5';
title2 'Using Path Model Syntax';

data torus;
  infile 'timmy1.data' firstobs=2;
  input id w1 w2 w3 w4 w5 w6 v1 v2 v3 v4;

ods exclude Calis.ML.SqMultCorr (persist);

proc calis pshort nostand;
  var w1 w2 w3 w4 w5 w6 v1 v2 v3 v4;
  path
    F_BrAw F_AdAw F_Inter --> F_PI = gamma1-gamma3,
    F_Inter F_PI --> F_PBeh = gamma4 beta,
    F_BrAw --> w1 = 1, F_BrAw --> w2 = lambda2,
    F_AdAw --> w3 = 1, F_AdAw --> w4 = lambda4,
    F_Inter --> w5 = 1, F_Inter --> w6 = lambda6,
    F_PI --> v1 = 1, F_PI --> v2 = lambda8,
    F_PBeh --> v3 = 1, F_PBeh --> v4 = lambda10;
  /* pvar works like the lineq's var statement. Naming an endogenous
  variable actually refers to the error term. /
  pcov F_BrAw F_AdAw F_Inter;
  /* Could name covariances as in lineq's cov statement. Again,
  names of endogenous variables actually refer to the
  corresponding error terms. */

```

Doughnut Shop Brand Awareness Study: Model 5 1
 Using Path Model Syntax

The CALIS Procedure
 Covariance Structure Analysis: Model and Initial Values

Modeling Information

Data Set	WORK.TORUS
N Records Read	200
N Records Used	200
N Obs	200
Model Type	PATH
Analysis	Covariances

Doughnut Shop Brand Awareness Study: Model 5 2
 Using Path Model Syntax

The CALIS Procedure
 Covariance Structure Analysis: Optimization

Levenberg-Marquardt Optimization

Scaling Update of More (1978)

Parameter Estimates	28
Functions (Observations)	55

Optimization Start

Active Constraints	0	Objective Function	0.1016760046
Max Abs Gradient Element	0.0313316908	Radius	1

Iter	Rest arts	Func Calls	Act Con	Objective Function	Obj Fun Change	Max Abs Gradient Element	Lambda	Actual Over Pred Change
1	0	4	0	0.09503	0.00665	0.00853	0	0.928
2	0	6	0	0.09482	0.000210	0.000876	0	0.932
3	0	8	0	0.09481	3.935E-6	0.000242	0	0.795
4	0	10	0	0.09481	2.198E-7	0.000052	0	0.752
5	0	12	0	0.09481	1.484E-8	0.000014	0	0.737
6	0	14	0	0.09481	1.062E-9	3.543E-6	0	0.732

Optimization Results

Iterations	6	Function Calls	17
Jacobian Calls	8	Active Constraints	0
Objective Function	0.0948118648	Max Abs Gradient Element	3.5432106E-6
Lambda	0	Actual Over Pred Change	0.7317126267
Radius	0.000356371		

Convergence criterion (ABSGCONV=0.00001) satisfied.

Doughnut Shop Brand Awareness Study: Model 5
Using Path Model Syntax

3

The CALIS Procedure
Covariance Structure Analysis: Maximum Likelihood Estimation

Fit Summary

Modeling Info	N Observations	200
	N Variables	10
	N Moments	55
	N Parameters	28
	N Active Constraints	0
	Baseline Model Function Value	4.9657
	Baseline Model Chi-Square	988.1793
	Baseline Model Chi-Square DF	45
	Pr > Baseline Model Chi-Square	<.0001
	Absolute Index	Fit Function
Chi-Square		18.8676
Chi-Square DF		27
Pr > Chi-Square		0.8748
Z-Test of Wilson & Hilferty		-1.1505
Hoelter Critical N		424
Root Mean Square Residual (RMSR)		0.4214
Standardized RMSR (SRMSR)		0.0195
Parsimony Index	Goodness of Fit Index (GFI)	0.9822
	Adjusted GFI (AGFI)	0.9638
	Parsimonious GFI	0.5893
	RMSEA Estimate	0.0000
	RMSEA Lower 90% Confidence Limit	0.0000
	RMSEA Upper 90% Confidence Limit	0.0286
	Probability of Close Fit	0.9939
	ECVI Estimate	0.3927
	ECVI Lower 90% Confidence Limit	0.4415
	ECVI Upper 90% Confidence Limit	0.4561
Akaike Information Criterion	74.8676	
Bozdogan CAIC	195.2204	
Schwarz Bayesian Criterion	167.2204	
McDonald Centrality	1.0205	

The CALIS Procedure
Covariance Structure Analysis: Maximum Likelihood Estimation

PATH List

-----Path-----	Parameter	Estimate	Standard Error	t Value
F_BrAw ----> F_PI	gamma1	0.22946	0.14548	1.57719
F_AdAw ----> F_PI	gamma2	0.36871	0.16173	2.27976
F_Inter ----> F_PI	gamma3	0.55314	0.17048	3.24455
F_Inter ----> F_PBeh	gamma4	-0.12922	0.25793	-0.50096
F_PI ----> F_PBeh	beta	0.54576	0.22445	2.43149
F_BrAw ----> w1		1.00000		
F_BrAw ----> w2	lambda2	0.52787	0.07713	6.84418
F_AdAw ----> w3		1.00000		
F_AdAw ----> w4	lambda4	0.54312	0.09055	5.99792
F_Inter ----> w5		1.00000		
F_Inter ----> w6	lambda6	1.09174	0.08091	13.49368
F_PI ----> v1		1.00000		
F_PI ----> v2	lambda8	0.70694	0.06596	10.71760
F_PBeh ----> v3		1.00000		
F_PBeh ----> v4	lambda10	1.03974	0.11022	9.43305

Variance Parameters

Variance Type	Variable	Parameter	Estimate	Standard Error	t Value
Exogenous	F_BrAw	_Add01	19.29669	3.13352	6.15814
	F_AdAw	_Add02	15.99021	3.05552	5.23321
	F_Inter	_Add03	15.03655	2.16915	6.93199
Error	w1	_Add04	5.06590	2.09037	2.42345
	w2	_Add05	13.03873	1.42404	9.15619
	w3	_Add06	7.07288	2.23517	3.16436
	w4	_Add07	13.46712	1.48786	9.05136
	w5	_Add08	6.25568	0.96719	6.46789
	w6	_Add09	6.12839	1.07124	5.72085
	v1	_Add10	8.32179	1.49034	5.58381
	v2	_Add11	10.35070	1.22403	8.45626
	v3	_Add12	4.63480	1.69444	2.73529
	v4	_Add13	3.82859	1.80292	2.12355
	F_PI	_Add14	3.31743	1.35011	2.45715
	F_PBeh	_Add15	12.68309	2.11239	6.00413

Covariances Among Exogenous Variables

Var1	Var2	Parameter	Estimate	Standard Error	t Value
F_AdAw	F_BrAw	_Add16	12.36235	1.87827	6.58177
F_Inter	F_BrAw	_Add17	13.54737	1.84523	7.34185
F_Inter	F_AdAw	_Add18	11.36898	1.70653	6.66205